

{overall conditions}
p1=10000
p9=10
T1=500
eta_s=0.80
h1=enthalpy(STEAM_NBS,T=T1,p=p1)
s1=entropy(STEAM_NBS,T=T1,p=p1)
s9ii=s1
h9ii=enthalpy(STEAM_NBS,s=s9ii,p=p9)

{We don't know what dhi is but we assume it is the same in all stages. You might have to start EES with a guess value of (h1-h9ii)/8}

{stage 1}
s2i=s1
h2i=enthalpy(STEAM_NBS,s=s2i,p=p2)
h2=h1-(dhi)*eta_s
s2=entropy(STEAM_NBS,h=h2,p=p2)
h1-h2i=dhi

{stage 2}

```
s3i=s2
h3i=enthalpy(STEAM_NBS,s=s3i,p=p3)
h3=h2-(dhi)*eta_s
s3=entropy(STEAM_NBS,h=h3,p=p3)
h2-h3i=dhi

{stage 3}
s4i=s3
h4i=enthalpy(STEAM_NBS,s=s4i,p=p4)
h4=h3-(dhi)*eta_s
s4=entropy(STEAM_NBS,h=h4,p=p4)
h3-h4i=dhi

{stage 4}
s5i=s4
h5i=enthalpy(STEAM_NBS,s=s5i,p=p5)
h5=h4-(dhi)*eta_s
s5=entropy(STEAM_NBS,h=h5,p=p5)
h4-h5i=dhi

{stage 5}
s6i=s5
h6i=enthalpy(STEAM_NBS,s=s6i,p=p6)
h6=h5-(dhi)*eta_s
s6=entropy(STEAM_NBS,h=h6,p=p6)
h5-h6i=dhi

{stage 6}
s7i=s6
h7i=enthalpy(STEAM_NBS,s=s7i,p=p7)
h7=h6-(dhi)*eta_s
s7=entropy(STEAM_NBS,h=h7,p=p7)
h6-h7i=dhi

{stage 7}
s8i=s7
h8i=enthalpy(STEAM_NBS,s=s8i,p=p8)
h8=h7-(dhi)*eta_s
s8=entropy(STEAM_NBS,h=h8,p=p8)
h7-h8i=dhi

{stage 8}
s9i=s8
h9i=enthalpy(STEAM_NBS,s=s9i,p=p9)
h9=h8-(dhi)*eta_s
s9=entropy(STEAM_NBS,h=h9,p=p9)
h8-h9i=dhi

Reheat=8*dhi/(h1-h9ii)
eta_t=(h1-h9)/(h1-H9ii)
eta_n=0.90
V2=(2*eta_n*dhi*K1)^.5
```

```
K1=1000 {J/kJ}
alpha=22
U=V2*(cos(alpha))/2
RPM=3600
K2=60 {s//min}
U=RPM/K2*pi*D
```

{Note that the overall efficiency is higher than the stage efficiency. This is due to internal reheat. Losses in one stage raise the enthalpy entering the next stage and make a little more work available to the next. On a process line it looks like there is reheat between stages. This doesn't mean that losses make things better.
Also note that the equal ideal enthalpy drop per stage leads to a roughly constant pressure ratio across the stages, not equal pressure drop which you would expect in incompressible flow.}

SOLUTION
Unit Settings: [kJ]/[C]/[kPa]/[kg]/[degrees]

α = 22.000 [deg]	**D = 1.360** [m]	dhi = 169.863 [kJ/kg]	η_n = 0.900
η_s = 0.800	η_t = 0.846 [-]	h1 = 3373.998 [kJ/kg]	h2 = 3238.108 [kJ/kg]
h2i = 3204.135 [kJ/kg]	h3 = 3102.218 [kJ/kg]	h3i = 3068.245 [kJ/kg]	h4 = 2966.327 [kJ/kg]
h4i = 2932.355 [kJ/kg]	h5 = 2830.437 [kJ/kg]	h5i = 2796.465 [kJ/kg]	h6 = 2694.547 [kJ/kg]
h6i = 2660.575 [kJ/kg]	h7 = 2558.657 [kJ/kg]	h7i = 2524.684 [kJ/kg]	h8 = 2422.767 [kJ/kg]
h8i = 2388.794 [kJ/kg]	h9 = 2286.876 [kJ/kg]	h9i = 2252.904 [kJ/kg]	h9ii = 2088.825 [kJ/kg]
K1 = 1000 [J/kJ]	K2 = 60 [s/min]	p1 = 10000.000 [kPa]	p2 = 5770.061 [kPa]
p3 = 3140.471 [kPa]	p4 = 1587.364 [kPa]	p5 = 727.860 [kPa]	p6 = 296.035 [kPa]
p7 = 108.092 [kPa]	p8 = 35.174 [kPa]	p9 = 10.000 [kPa]	**Reheat = 1.057** [-]
RPM = 3600.000 [1/min]	s1 = 6.597 [kJ/kg-K]	s2 = 6.646 [kJ/kg-K]	s2i = 6.597 [kJ/kg-K]
s3 = 6.702 [kJ/kg-K]	s3i = 6.646 [kJ/kg-k]	s4 = 6.765 [kJ/kg-K]	s4i = 6.702 [kJ/kg-K]
s5 = 6.839 [kJ/kg-K]	s5i = 6.765 [kJ/kg-K]	s6 = 6.923 [kJ/kg-K]	s6i = 6.839 [kJ/kg-K]
s7 = 7.013 [kJ/kg-K]	s7i = 6.923 [kJ/kg-K]	s8 = 7.112 [kJ/kg-K]	s8i = 7.013 [kJ/kg-K]
s9 = 7.218 [kJ/kg-K]	s9i = 7.112 [kJ/kg-K]	s9ii = 6.597 [kJ/kg-K]	T1 = 500.000 [C]
U = 256.343 [m/s]	V2 = 552.949 [m/s]		

No unit problems were detected.
Purple units were automatically set. Right click on the variable to confirm or change the units.

```
m_dot = rho*Q
m_dot=1e5/3600
rho=1000
p_in=10000
p_out=2000000
H=(p_out-p_in)/rho
omega=3600/60*2*pi
k_s=omega*Q^0.5/H^0.75
{ks turns out to be about 0.21 which is at the low end for a centrifugal pump}
sigma_pump =0.03 {from ks and figure 2.8}
sigma_application=1.5*sigma_pump
sigma_application=NPSH/H
NPSH=(p_in-p_v)/rho+g*S
{S is the difference in elevation between the condenser outlet and the pump inlet required to prevent
cavitation if the condenser outlet pressure and the vapor pressure at the pump inlet are equal,
 i.e., saturated liquid.}
p_v=10000
g=9.8
NPSH=(p_in)/rho+g*S_1
{S_1 is the difference in elevation between the condenser outlet and the pump inlet required to prevent
cavitation if the vapor pressure is reduced to near zero by subcooling, i.e., compressed liquid.}
{Note that in this case you don't gain much by subcooling.  Subcooling helps more when the initial
saturation pressure is higher}
```

SOLUTION

Unit Settings: [J]/[C]/[Pa]/[kg]/[degrees]

g = 9.8 [m/s^2]	H = 1990 [J/kg]	**k_s = 0.2109** [-]	$\dot{m}$ = 27.78 [kg/s]
$NPSH$ = 89.55 [J/kg]	ω = 377 [1/s]	p_{in} = 10000 [Pa]	p_{out} = 2.000E+06 [Pa]
p_v = 10000 [Pa]	Q = 0.02778 [m3/s]	ρ = 1000 [kg/m^3]	**S = 9.138** [m]
$\sigma_{application}$ = 0.045 [-]	**σ_{pump} = 0.03** [-]	**S_1 = 8.117** [m]	

No unit problems were detected.

```
rho=1000
dp_pump=13333-3333*Q^2
dp_sys=7500+2500*Q^2
dp_pump=dp_sys
eta=0.8*(1-(Q-1)^2)
H=dp_pump/rho
P=rho*Q*H/eta
m_dot=rho*Q
Cp=4.2
Cv=Cp
dT=H/Cv*(1-eta)/eta
{dT=5}
```

SOLUTION

Unit Settings: [kJ]/[C]/[kPa]/[kg]/[degrees]

Cp = 4.2 [kJ/kg-K]	Cv = 4.2 [kJ/kg-K]	dp_{pump} = 10000 [kPa]	dp_{sys} = 10000 [kPa]	**dT = 0.595** [C]
η = 0.800 [-]	**H = 10.00** [kJ/kg]	$\dot{m}$ = 1000 [kg/s]	**P = 12500** [kW]	**Q = 1.000** [m^3/s]
ρ = 1000 [kg/m^3]				

No unit problems were detected.

```
rho=1000
dp_pump=13333-3333*Q^2
{dp_sys=7500+2500*Q^2}
{dp_pump=dp_sys}
eta=0.8*(1-(Q-1)^2)
H=dp_pump/rho
P=rho*Q*H/eta
m_dot=rho*Q
Cp=4.2
Cv=Cp
dT=H/Cv*(1-eta)/eta
dT=5
```

SOLUTION

Unit Settings: [kJ]/[C]/[kPa]/[kg]/[degrees]

Cp = 4.2 [kJ/kg-K]	Cv = 4.2 [kJ/kg-K]	dp_{pump} = 13074 [kPa]	**dT = 5.000** [C]	η = 0.384 [-]
H = 13.07 [kJ/kg]	$\dot{m}$ = 278.6 [kg/s]	**P = 9494** [kW]	**Q = 0.279** [m^3/s]	ρ = 1000 [kg/m^3]

No unit problems were detected.